

Compositionality in embedded DSLs

(talk proposal)

Jeremy Gibbons

Department of Computer Science, University of Oxford
<http://www.cs.ox.ac.uk/people/jeremy.gibbons/>

1. Context

There are two main approaches to implementing domain-specific languages. With the *standalone* approach [1, 5], independent tools such as compilers and run-time environments for the DSL are implemented in one or more general-purpose programming languages. With the *embedded* approach [3, 8], the DSL implementation takes the form of a library of definitions in the host language, and a program in the DSL is merely a program in the host language that makes use of the library.

Amongst embedded DSLs, there are two further refinements. With a *deep embedding*, terms in the DSL are implemented simply to construct an abstract syntax tree (AST), which is subsequently transformed for optimization and traversed for evaluation. With a *shallow embedding*, terms in the DSL are implemented directly by their semantics, bypassing the intermediate AST and its traversal. Deep embeddings might seem like the obvious approach, but Kamin [10] and Erwig [4] (among others) argue that shallow embeddings are superior.

Our focus in this talk proposal is the relationship between deep and shallow embeddings of DSLs, and the connection to compositional semantics.

2. Embeddings

Consider for example a DSL for 2D graphics, which might involve sublanguages for affine transformations (translation, scaling, rotation). As a deep embedding, this sublanguage could be represented using an algebraic datatype:

```
data Transform where
  Identity  :: Transform
  Translate :: Complex → Transform
  Scale     :: Real → Transform
  Rotate    :: Real → Transform
  Compose  :: (Transform, Transform) → Transform
```

(The notation is Haskell; we trust that it is sufficiently self-explanatory.) This sublanguage might be used in various ways within the overall graphics language. For example, we probably want to transform points:

```
transform :: (Transform, Complex) → Complex
transform (Identity, p)      = p
transform (Translate p, q)   = p + q
transform (Scale s, p)       = scale s p
transform (Rotate a, p)      = rotate a p
transform (Compose (t,u), p) = transform (t, transform (u, p))
scale, rotate :: Real → Complex → Complex
scale s p = (s:+0) × p
rotate a p = mkPolar (magnitude p) (a + phase p)
```

With a shallow embedding, we dispense with the abstract syntax trees (that is, the algebraic datatype), and represent terms directly by their semantics. This is straightforward to do for a single interpretation, for example representing a transformation directly as a function on points:

```
type TransformS = Complex → Complex
identityS :: TransformS
identityS    = λp → p
translateS :: Complex → TransformS
translateS p = λq → p + q
scaleS :: Real → TransformS
scaleS s     = λp → scale s p
rotateS :: Real → TransformS
rotateS a    = λp → rotate a p
composeS :: (TransformS, TransformS) → TransformS
composeS (f, g) = f ∘ g
```

Note the similarity between the interpretation of a deep embedding (such as the clauses of *transform*) and the direct representation in a shallow embedding (via *identity_S*, *translate_S*, etc).

3. Tension

With a deep embedding, it is trivial to provide additional interpretations of a language:

```
isLinear :: Transform → Bool
print    :: Transform → String
```

But what about multiple interpretations in a shallow embedding? We don't want to have to redefine the representation *Transform* and reimplement all the constructors for each new interpretation. Sometimes we are lucky enough to have a common generalization of multiple interpretations (for example, we could represent a *Transform* as a matrix, and implement *transform* and *isLinear* in terms of this) but we are not always so lucky (it doesn't work for *print*). What is the general solution?

4. Resolution

The general pattern is that

each feasible shallow embedding of a language corresponds to a *compositional* interpretation of the deep embedding of the language in question.

For example, the function *transform*—or rather, its curried version *transform'* :: *Transform* → (*Complex* → *Complex*)—is compositional, in the sense that the interpretation *transform'* (*Compose* (*t*, *u*)) of a term *Compose* (*t*, *u*) depends only on the interpretations *transform'* *t* and *transform'* *u* of its subterms *t* and *u*, and not on any

other attributes of t and u . This is both a necessary and a sufficient condition for $transform'$ to be feasible as a shallow embedding of the language of transformations.

In functional programming, such compositional functions are called *folds* [9]. Folds follow a fixed recursion pattern, corresponding to the shape of the data structure; their points of variation constitute what is called an *algebra*, specifying how to interpret each constructor of the datatype. For example, the *Transform* datatype has five constructors, so an algebra for this shape of data is a quintuple, with one component per constructor:

type $TAlg\ a = (a, Complex \rightarrow a, Real \rightarrow a, Real \rightarrow a, (a, a) \rightarrow a)$

The fold operator for *Transforms* takes such an algebra, and collapses a *Transform* down to a value; the recursion follows the shape of the *Transform*, and the individual constructors are handled by the corresponding components of the algebra:

```
fold :: TAlg a → (Transform → a)
fold (i,t,s,r,c) Identity      = i
fold (i,t,s,r,c) (Translate p) = t p
fold (i,t,s,r,c) (Scale a)    = s a
fold (i,t,s,r,c) (Rotate a)   = r a
fold (i,t,s,r,c) (Compose (f,g)) = c (fold (i,t,s,r,c) f,
                                       fold (i,t,s,r,c) g)
```

Compositional interpretations can be expressed as folds; for example,

$transform' t = fold (id, (+), scale, rotate, uncurry (\circ)) t$

One can see folds as the essence of compositional interpretations. And this gives us a clue about supporting multiple interpretations in a shallow embedding: if interpretations in a shallow embedding have to be compositional, and compositional interpretations are all and only those expressible as folds, then

the fold pattern is precisely the least common generalization of all shallow interpretations.

The folding pattern is what all shallow interpretations have in common; the instantiation of the pattern—that is, the specific algebra—is what varies. Consider a version of *fold* with its arguments in the opposite order:

$flip\ fold :: Transform \rightarrow (TAlg\ a \rightarrow a)$

We should use the result type $TAlg\ a \rightarrow a$ of this function as the semantic domain for our *parametrized* shallow embedding; it can then be instantiated to any fold by supplying the corresponding algebra.

type $Transform_A = \forall a. TAlg\ a \rightarrow a$

(For technical reasons, the type parameter a above has to be explicitly quantified rather than left unbound.) All the constructors of the language can be implemented easily under this representation:

```
identity_A :: Transform_A
identity_A (i,t,s,r,c) = i
translate_A :: Complex → Transform_A
translate_A p (i,t,s,r,c) = t p
scale_A :: Real → Transform_A
scale_A a (i,t,s,r,c) = s a
rotate_A :: Real → Transform_A
rotate_A a (i,t,s,r,c) = r a
compose_A :: (Transform_A, Transform_A) → Transform_A
compose_A (f,g) (i,t,s,r,c) = c (f (i,t,s,r,c), g (i,t,s,r,c))
```

And any compositional interpretation arises from applying the shallow embedding (which is a fold computation) to the appropriate algebra:

$transform_A :: Transform_A \rightarrow (Complex \rightarrow Complex)$
 $transform_A t = t (id, (+), scale, rotate, uncurry (\circ))$

Many seemingly non-compositional interpretations are still expressible as folds, if looked at in the right way. For example, the interpretation *isLinear* above is non-compositional, because to determine whether $Compose\ (t, u)$ is linear, it does not suffice to know whether t and u are linear. Still, it is a simple projection from *transform*, which is compositional:

$isLinear\ t = (transform\ (t, 0) == 0)$

Mutually dependent interpretations can be defined together as a pair. Context-dependent interpretations, such as precedence-aware printing, can be turned into context-independent compositional higher-order interpretations:

$printPrec :: Transform \rightarrow (Precedence \rightarrow String)$

5. Conclusion

Deep and shallow embeddings are more popular in functional programming circles than in object-oriented ones. That's not so surprising, give as we have seen that they depend heavily on algebraic datatypes and higher-order functions, respectively. Still, modern language design (C#, Scala, Python) combines the best of both paradigms, so hopefully that barrier will gradually recede. Then the lightweight embedded approach will become more widely available.

6. Acknowledgements

This talk proposal is based on a functional pearl [7] appearing at ICFP 2014, joint work with Nicolas Wu, which in turn grew out of lectures [6] given at the Central European Functional Programming summer school in 2013. The work was supported by EPSRC research grant EP/J010995/1 on *Unifying Theories of Generic Programming*.

References

- [1] J. Bentley. Programming pearls: Little languages. *Comm. ACM*, 29(8):711–721, 1986. Also Chapter 9 of [2].
- [2] J. Bentley. *More Programming Pearls: Confessions of a Coder*. Addison-Wesley, 1988.
- [3] R. Boulton, A. Gordon, M. Gordon, J. Harrison, J. Herbert, and J. V. Tassel. Experience with embedding hardware description languages in HOL. In *Theorem Provers in Circuit Design*, pages 129–156. North-Holland/Elsevier, 1992.
- [4] M. Erwig and E. Walkingshaw. Semantics-driven DSL design. In M. Mernik, editor, *Formal and Practical Aspects of Domain-Specific Languages: Recent Developments*, pages 56–80. IGI-Global, 2012.
- [5] M. Fowler. *Domain-Specific Languages*. Addison-Wesley, 2011.
- [6] J. Gibbons. Functional programming for domain-specific languages. In V. Zsóok, editor, *Central European Functional Programming Summer School*, volume 8606 of *LNCS*, pages 1–27. Springer, 2014. To appear.
- [7] J. Gibbons and N. Wu. Folding domain-specific languages: Deep and shallow embeddings. In *ICFP*, Sept. 2014.
- [8] P. Hudak. Building domain-specific embedded languages. *Comput. Surveys*, 28(4), 1996.
- [9] G. Hutton. A tutorial on the universality and expressiveness of fold. *J. Funct. Prog.*, 9(4), 1999.
- [10] S. M. Kamin and D. Hyatt. A special-purpose language for picture-drawing. In *Domain-Specific Languages*. Usenix, 1997.